;/* Variables used in this routine */


;


; All buffer 	= $0488 - $0524 = $9C bytes


; Final buffer 	= $0488 - $04C7


; Work buffer	= $04C8 - $0508


; Free Variables= $051A - $0524


;  


; NOTE: Bottom of Font $0100 away from TOP in VRAM


;


; Current Tile 	= $0480


; Current ROW	= $0482


; Vram pointer	= $0484 AND Stack


; Buffer Pointer= $0486


; Font character= $0C-$0D


; Routine Start	= $C02966 ( 0x002B66 )


; Routine MAX	= $C02C04 ( 0x002E04 )


; 


;


;


.HIROM ON


.ORG $000000		





;##########################


; Variables that exist    #


;##########################


.Table		$80


 tile	.dsb 2	; char per line counter


 row	.dsb 2	; current row text is 


 Vram	.dsb 2	; Vram pointer


 bptr	.dsb 2 	; font buffer pointer


 fmem	.dsb 64	; font buffer -> DMA is transfered from here





 bits	.dsb 2	; bits left on current tile


 cntr	.dsb 2	; universal counter


 xpos	.dsb 2	; X register font data position


 tmp1	.dsb 2	; for anything


 tmp2	.dsb 2	; for anything


 wmem	.dsb 64	; working memory -> No background data stored here


.end





 char	.equ $000c


;##########################


; Variables that exist    #


;##########################











;#######################


; Start of all code... #


;####################### 


Main_Proc:		; Assumes Direct Page = $0400


 rep #$30		; 16 - bit Acc, X, and Y





 lda tile


 asl : asl


 asl : asl		; multiply tile pointer by 16 ( READ FORMULA AT END OF FILE )


 adc #!$3000	; text data VRAM start position


 sta Vram		; store data in vram





 lda !char		; load the characters's value


 and #!$00ff	; clearall high byte data


 sta char		; save as new font character


 asl : asl		; multiply font value by 16 ( READ FORMULA AT END OF FILE )


 asl : asl


 sta xpos		; store it as the ROM pointer	





 rep #$20		


 ldx xpos		; load X with font pointer


 ldy #$0000		; clear Y





 sep #$20


 stz bits+1


 stz cntr+1


 stz tmp1+1








;#####################################


; Start loading the font and adjust  #


; where needed. Simple routine.      #


;#####################################


VWF_routine:  


 lda bits		; if there are bits left, then we must use


 bne _NoClear_	; the data on the current tile





 jsr _ClearBuf	; clear all of working memory





;######################################


; Start data transfer for font goodies#


;######################################


_NoClear_:		; New tile or VWF tile


 pha			; if we have bits left or not, save the bits left for later math


 lda #$08		; load a tile's width


 sec


 sbc $01,s		; subtract the bits left on the current tile from a tile width


 pha			; save this value as the shift amount for the new font


 lda #$10		; 16-loops per char 


 sta cntr		; font load counter





;####################


; Adjust the tiles  #


;####################


Adjust:


 sep #$20


 lda $01,s		; load up the shifting amount


 and #$07		; if the value is >= 8, clear it and save it for a future test


 sta bits


			; 00110011 00000000 >> 2 = 00001100 11000000


 lda font,x		; load a piece of data


 xba 			; set it as the high byte


 lda #$00 		; load a clear byte to catch shifted bits


 inx			; zero is loaded so that SHIFTED bits aren't lost 


 rep #$20		; shifted bits are used for the VWF routine


 


 inc bits 		; Check to see if it's a NEW tile or a VWF tile


 dec bits		; add 1 then subtract 1. If it's ZERO then do NO shifting


 beq _SkipShift  	; if bits = 0, then it's a NEW tile





;##########################################


; Only come here when a font needs to be  #


; shifted over to fit in the same tile    #


; the last font is stored into.           # 


;##########################################


Shift:		; shifting was needed


 lsr			; shift the new font over by the number of pixels


 dec bits		; left on the current working tile


 bne Shift 		; loop until it's done








;###########################


; Just come here and store #


; the font data when it's a#


; regular NON-shifted font #


;###########################


_SkipShift:		; no shifting needed


 jsr _Add_Data	; add in the NEW font tile data ( READ THE END OF FILE FOR DETAILS )


 dec cntr


 bne Adjust





 jsr TransferFont	; tranfer font from Working buffer to DMA offset buffer


 sep #$20


 lda $02,s		; load the amount of bits left on the current tile


 ldx !char		; load the current working character value


 sec			


 sbc wdth,x		; subtract the current working value from the bits left


 bpl NoOverFlow	; if the value is >= 0 then it fits on the current tile








;##########################


; Current font overflowed #


;##########################


; DETAILS. say... "a tile size = 8 pixels"


; bits left = 4 pixels, 4 pixels are used already


; "new font size = 6 pixels"


; so 4 - 6 = - 2 ( $FE ), that is less than ZERO


; and means all pixels were use on the current font and we need to use 2 pixels from the new tile


;


OverFlow:


 eor #$ff		; this makes $FE = '1' 


 inc			; add one ALWAYS because of the negative value is actually


 sta tmp1		; 1 less than what we need. ZERO is between $FF and $01, said add 1 for the ZERO


 lda #$08		; load a tile size


 sec			


 sbc tmp1		; subtract that value we just got from the subtraction from the tile size


			; that value is the amount of pixels used on a NEW tile


 and #$07		; clear any bits greater than 7


 sta bits		; ZERO means a complete tile was used again


 inc tile		; goto the next tile to store new data on


			; ( VRAM is already set, so Adding one is ok now )


 lda $02,s		; safety purpose for NULL bits


 beq _NoDec		; if NULL bits, it was a non-shifted tile








;#######################


; Transfer working data#


; only for next font.. #


;#######################


 ldx #$0000


_Move_:	   	; ONLY for an overflow, move data in the WORKING buffer ONLY, not the DMA buffer


 lda wmem+32,x	; load the over flowed bits ( EXPLANATION is at the ENF OF FILE )	


 sta wmem+00,x	; store it at the begining


 stz wmem+32,x	; clear the overflowed data


 inx


 cpx #$001f


 bne _Move_


 bra _NoDec








;########################################


; It was a good tile size, no more data #


; needed to be altered or what-not	#


;########################################


; Say "bits left = 4


; and new font was only 2 pixels


; 4 - 2 = 2, so that is NO overflow


; clear any value greater than 7


; and save this value as the new bits left on current tile


;


;


NoOverFlow:		; assumes NEW bits left in Accum


 and #$07


 sta bits


 bne _NoDec		; if "new font was 4 bits" and "bits left = 4", then current tile is finished


 inc tile		; if the tile is finished, goto next tile





_NoDec:


 plx			; pull garbage, bits left and the shift counter are on stack, both are BYTE size data


			; so pull a WORD size X value





Done_Font:


 sep #$20


 jmp DrawIt_or_Whatever	; this is game dependent


;/* END OF MAIN */











;#############


; 8x16 data  #


;#############


font:				; font and width table storage area


 .INCBIN "CUSTOM.BIN"


Wdth:			


 .INCBIN "CUSTOM.WID"











;#############################


;			   	     #


; Notes on things from above #


;			           #


;#############################


;


; VRAM Set-up formula:


; if ( Font is 8 pixels wide ) then


;   Vram = Tile * 16 + BaseOffset -> ( Vram is WORD size - 2 bytes per offset )


;   if ( Tile is store - Fig 1 ) Vram = Tile * 32 + BaseOffset


; else


; if ( Font is 16 pixels wide ) then


;   Vram = Tile * 32( or 64 Fig 1) + BaseOffset -> ( Vram is WORD size - 2 bytes per offset )


;


;Fig 1:


;�	<- image is the letters "AB" in Top-Bottom format


;


;


; Font Set-up formula:


; if ( Font is 8 pixels wide ) then


;   Xpos = Char * 8 + BaseOffset -> 8x8 


;   if ( Font 8x16 ) Xpos = Char * 16 + BaseOffset


; else


; if ( Font is 16 pixels wide ) then


;   Xpos = Char * 32 + BaseOffset -> ( Vram is WORD size - 2 bytes per offset )


;   if ( Font 16x16 ) Xpos = Char * 32 + BaseOffset


;


;Fig 2:


;�	<- image is the letters "AB" in Top-Bottom format


;


;


;


;


;Setting up the Fonts on the Tiles:


;First Example is the NO-Overflow Example:


; here is "l i"


; top pic is current tile and new font ( "i" is the new font )


; bits left = 5 since the width of "l" + 1 pixel space = 3  


; Since 3 pixels are used, we must shift the "i" over by the size of "l" + space


; we get the value like so: ShiftAmount = (8 - Bits Left)


; so we shift "i" over and we get the middle pic


; Now we must add the shifted "i" to "l". That's done like so by OR'ing NOT ADDING...


; 11000000 


; 00011000 


; --------


; 11011000


; 


; That gives us the Bottom pic, our Variable Width Font!


; To get the Bits Left, we do this...


; Bits Left = ( Bits Left - Width_of["i"] )


; in REAL numbers...


; Bits Left = 5 - 3


;


;�


;


;


;


;Setting up the Fonts on the Tiles:


;Second Example is the Overflow Example:


; here is "f o"


; top pic is current tile and new font ( "o" is the new font )


; bits left = 1 since the width of "f" + 1 pixel space = 7  


; Since 7 pixels are used, we must shift the "o" over by the size of "f" + space


; we get the value like so: ShiftAmount = (8 - Bits Left)


; so we shift "o" over and we get the middle pic


; Now we must add the shifted "o" to "f". 


; Notice that ONLY the first pixel line in "o" was saved when shifted.


; That means we have an overflow!!!


; That's where this comes into play...


; lda font,x 		; load a piece of data


; xba 			; set it as the high byte


; lda #$00 			; load a clear byte to catch shifted bits


; the rest of "o" is in the clear byte you loaded as the low byte of the Accumulator


; that byte was store 32 bytes away from the current working buffer tile


; NOTE: the working buffer tiles look like the pictures we are looking at now


; So you do the samething that was done with the "l i" but instead of DMA'ing 1 tile, you DMA


; 2 tiles. That's why the buffers are 64 bytes in size ( 2 - 8x16 2-Bpp tiles )


;


;Transfering OVERFLOWED data:


; after the data is transfered to the screen, transfer the OVERFLOWED part of "o" to 


; the begining of the working buffer and repeat the process all over again.


; remember to reset the bits left for the NEW tile.


; Do it like this... [ I will use REAL numbers for better understanding. "f" = 7, "o" = 8 ]


; assuming the "f" is stored and bits left now = 1 pixel remaining


;


; Actual Formula:


; Bits Left = TileSize - ((( BitsLeft - Width_Of["o"] ) EOR $FF) + 1)


;


; Bits Left = 8 - ((( 1 - 8 ) EOR $FF) + 1)


; 1 - 8 = $F9 (-7)


; $F9 EOR $FF = 6


; 6 + 1 = 7


; 8 - 7 = 1


; New Bits Left = 1


; That's it!


;


;�





